

PRINCIPES D'UTILISATION DES CARTES  
D'EXTENSION-RAM vortex SP64 - SP512

La memoire vive de l'extension RAM vortex (qui sera appelee par la suite "RAM externe") peut etre directement adressee par le Z80 de votre CPC. Toutefois le Z80 ayant, comme la plupart des microprocesseurs 8 bits, un bus d'adresses de 16 bits, la totalite de la memoire disponible (de 128 a 576 Ko) ne pourra pas etre adressee autrement que par pages de 64 Ko ou moins.

Le principe consistant a diviser la RAM en pages et a rendre ces pages adressables par un microprocesseur donne est appele "mapping" ou encore "bank-switching". Il doit etre utilise a chaque fois que la capacite d'adressage du microprocesseur est inferieure a la totalite de la memoire supportee par le systeme. Pour les CPC ce principe ne represente rien de nouveau dans la mesure ou, meme non equipes de RAM externe, ils effectuent en permanence du bank-switching. Par exemple, la ROM BASIC et la RAM video (memoire ecran) occupent des zones de memoire qui, pour le Z80, sont logees a la meme adresse.

Il est donc impossible de les adresser simultanement car ceci creerait des conflits au niveau du bus de donnees, et ces divers acces sont controles par le gate array.

L'adjonction de signaux de controle et l'utilisation de ce qu'on appelle un "port de configuration" (adresse 7Fxx) permettent de faire en sorte que, a un instant donne, une seule des memoires logees a la meme adresse soit accessible (par exemple, soit la ROM BASIC, soit la RAM video).

Il est aise de comprendre que la presence de seulement 16 bits pour les adresses est insuffisante pour selectionner une adresse precise dans une page memoire donnee.

La RAM externe vortex peut etre maniee par blocs de 32 Ko qui peuvent etre commutes a l'interieur ou hors de l'espace adresse du Z80 (voir figures). Le controle de cette commutation est exerce par 2 ports :

- Selection de banque/accessibilite a la RAM externe: port FBBD
- Selection de bloc/accessibilite du bloc selectionne: port 7Fxx

```
-----
* Definitions : *
* ----- *
* * *
* Banque : une zone de 64 Ko contigus qui occupe la totalite *
* de l'espace adresse du Z80 (de 0000H a FFFFH). *
* Une telle zone peut etre rendue accessible ou *
* inaccessible au microprocesseur. *
* * *
* Bloc : une zone constituee de 32 Ko. Chaque banque est *
* constituee de 2 blocs : *
* - un bloc inferieur ( de 0000 a 7FFF ) *
* - un bloc superieur ( de 8000 a FFFF ) *
* * *
-----
```

1 - Bankselect: FBBD (adresse du port):  
-----

La RAM externe possede un registre de selection de banque dans lequel il est uniquement possible d'ecrire. Ce registre n'a aucun rapport avec les operations de lecture/ecriture en memoire. Il est seulement utilise pour commuter des banques entre elles, c'est a dire pour les rendre adressables par le Z80.

De plus, ce registre contient aussi l'interrupteur qui permet d'activer l'integralite de l'extension. Les differents bits de ce registre ont la signification suivante :

BIT	ACTION
0-2	Selection de banque. Permet de choisir une des 8 banques de 64 Ko pour les operations RAM qui vont suivre. Ces 3 bits permettent d'etendre virtuellement la capacite d'adressage du Z80 jusqu'a 19 bits (=512 kO).  000 --> banque 0    001 --> banque 1    010--> banque 2 etc.
3	Extension RAM - desactivation de ROM 0 --> ROM activee 1 --> ROM desactivee
4	Desactivation RAM. La mise a 1 de ce bit desactive toutes les RAM (autant les RAM externes que la RAM interne du CPC), et l'acquisition de donnees n'est plus possible que depuis les ROM. Avant d'utiliser ce bit, il est necessaire de s'assurer que la suite du programme ne fera pas appel a une des RAM (attention, l'instruction CALL,nnnn utilise la pile de la RAM). Ce bit ne doit normalement pas etre utilise. Son usage est reserve pour des applications ulterieures.
5	Extension RAM - Desactivation de RAM 0 --> RAM desactivee 1 --> RAM activee Ce bit peut etre considere comme 'l'interrupteur principal' de la carte RAM. S'il n'est pas active, le "bit de mapping" du port 7Fxx (voir paragraphe suivant) sera sans effet, la carte d'extension n'etant pas accessible. Ce bit de controle a ete cree essentiellement a cause de certains logiciels sur cassette, particulierement des jeux, qui ont un chargeur special utilisant le port du gate array. La non desactivation de la carte RAM avant le chargement de ces jeux provoquera un blocage du systeme.

Normalement, seuls les bits 0, 1, 2 et 5 sont necessaires pour travailler avec la carte. Les bits 3 et 4 sont reserves pour des applications particulieres et bloqueront le systeme s'ils ne sont pas utilises correctement.

2 - Blockselect/ Mapping Enable: 7Fxx  
-----

Le registre BC doit etre exclusivement utilise pour la selection des blocs superieur et inferieur.

Le port 7Fxx est utilise par le gate array et sert a plusieurs

operations (selection des couleurs, du mode d'affichage ecran, activation/desactivation des ROM ).  
Ce dernier exemple montre que le gate array prend en charge certaines fonctions propres au mapping. Certains bits du port 7Fxx ne sont pas utilises par le gate array ( voir le COMPLETE FIRMWARE d'AMSTRAD ) et serviront donc au controle de la RAM externe. Ces bits sont les suivants :

BIT	ACTION
6	Blockselect - Selectionner le bloc inferieur (0) ou le bloc superieur (1) de la banque activee.
5	Mapping Enable - La mise a 1 de ce bit commute le bloc selectionne par le bit 6 dans la banque activee par les bits 0 a 2 du port FBBD dans l'espace adressable par le Z80. Le bloc correspondant de la RAM interne est desactive.

Cas particulier: Si le bit 7 est a 1, le bit 6 a 0 et le bit 5 a 1, alors le mot de commande est evalue par le Gate-Array et le bloc inferieur est selectionne (ce qui est indispensable lors de la gestion des interruptions). Cette technique est utilisee avec le BOS, pour etendre l'espace programmable du BASIC.

#### Les interruptions et l'extension RAM vortex:

Definition: le Z80 possede un signal d'entree qui lui permet d'interrompre un programme en cours d'execution pour se brancher a une adresse memoire donnee.

Pour tous les CPC, les fonctions de controle essentielles sont basees sur les interruptions. S'il n'y a pas d'autre peripherique, la seule source d'interruption est le gate array. Ce dernier necessite une interruption du Z80 300 fois par seconde. Dans le cas ou le firmware se contenterait de reagir aux interruptions sans analyse complementaire, la vitesse de l'ordinateur pourrait etre considerablement reduite, son microprocesseur etant principalement occupe a servir les interruptions. C'est pourquoi le systeme des CPC contient une liste d'evenements (ou event queue) avec des ordres de priorite. De cette facon, la diminution de vitesse du systeme est insignifiante (voir a ce sujet le COMPLETE FIRMWARE d'AMSTRAD ).

Le Z80 possede 3 modes d'interruption differents: les modes 0, 1 et 2. Le programmeur peut choisir n'importe lequel de ces modes avec l'instruction IM adequate.

Dans le CPC, le mode d'interruption du Z80 est le mode 1. C'est le plus contraignant des modes d'interruption du Z80 dans la mesure ou il effectue toujours un branchement a la meme adresse et ne permet pas de distinguer les sources d'interruption de facon directe. Cependant, il presente l'avantage de ne pas demander d'apport supplementaire de materiel.

Quand le Z80 du CPC accepte une interruption, il arrete le fonctionnement du programme en cours et se branche a l'adresse 0038 (hexadecimale). A cette adresse, il trouve le pointeur lui permettant d'executer la routine de traitement de l'interruption.

Ce pointeur ne doit pas être traité avec inattention, sinon le blocage du système est immédiat. Ceci est à considérer plus particulièrement quand l'utilisateur veut travailler lui-même avec l'extension, sans utiliser le software (BOS et le patch CP/M) qui est fourni avec la carte vortex. Par exemple les interruptions doivent être interdites (par l'instruction DI) à chaque fois qu'un bloc est commuté à l'intérieur ou hors de l'espace adressable par le Z80. De plus il faut s'assurer, lorsque les interruptions sont à nouveau autorisées (par l'instruction EI), que le pointeur de l'adresse 0038 est correctement renseigné. Dans le cas où l'extension est utilisée uniquement pour le stockage de données, le contrôle de la carte est relativement simple. C'est différent lorsque l'extension est utilisée pour étendre la mémoire programmable. Dans ce dernier cas, toute programmation doit être prudemment structurée et vérifiée afin de s'assurer que les interruptions seront traitées convenablement.

#### EXEMPLES

=====

Ci-dessous vous trouverez quelques exemples simples en langage machine, conçus comme programmes explicatifs sur l'utilisation de la carte vortex.

##### 1 Stockage de données dans un block inférieur (0000H-7FFFH):

Le programme de contrôle et la pile doivent (!) se trouver au dessus de 7FFFH, parce que le bloc inférieur est 'effacé' en activant la mémoire extérieure, alors les programmes et/ou la pile sont perdus et le système se plante.

Tâche: Copier un block de données du tampon TAMP vers le bloc inférieur de la banque 0 à l'adresse 0. Le tampon doit bien sûr se trouver au dessus de 8000H !

```

SELECT    EQU 0FBBDH      ; BANKSELECT PORT DE LA CARTE
BUFF      EQU 09000H      ; TAMPON DE DONNEES

                ORG 8000H      ; OU UNE ADRESSE DE DEBUT PLUS HAUT
                LD (SSTACK),SP ; SAUVEGARDE POINTEUR-PILE ET CREE UNE
                LD SP,SSTACK  ; PILE LOCALE AU DESSUS DE 8000H
;
; SELECTIONNE UNE BANQUE EXTERNE
;
                LD BC,SELECT   ; ADRESSE DU PORT DE 2 OCTETS POUR
                ; BANKSELECT
                LD A,00100000B ; SELECTIONNE BANQUE 0
                ; ACTIVE CARTE-RAM (BIT 5=1)
                OUT (C),A      ; ECRIRE DANS LE REGISTRE BANKSELECT DE
                ; LA CARTE
;
; Remarque: Jusqu'à la prochaine OUT sur la porte bankselect, la
; banque 0 est sélectionnée, c.a.d. qu'il n'est pas
; nécessaire de sélectionner la banque quand le numéro
; de banque ne change pas.
;
; SELECTIONNE LE BLOC INFÉRIEUR ET ACTIVE LA RAM EXTERNE
;
                DI            ; INTERDIT L'INTERRUPTION PENDANT LE
                ; TRANSFERT DE DONNEES
                EXX          ; SAUTE VERS LES REGISTRES ALTERNES

```

```

;.....
;
;   PUSH BC           ; SAUVEGARDE STATUS DU GATE-ARRAY
;   RES 6,C           ; SELECTIONNE BLOCK INFERIEURE
;   SET 5,C           ; ACTIVE RAM-EXTERNE
;   OUT (C),C         ;
;
;
; MAINTENANT LA RAM-EXTERNE EST ACTIVEE, LES DONNEES PEUVENT ETRE
; TRANSFEREES
;
;
;   LD HL,TAMP        ; POINTEUR VERS TAMPON
;   LD DE,0           ; ADRESSE DESTINATION DANS BANQUE 0
;   LD BC,128         ; TRANSFERER P.EX. 128 OCTETS
;   LDIR              ; TRANSFERE BLOC
;.....
;
; PUIS DESACTIVE LA RAM EXTERNE
;
;   POP BC            ; CHERCHE STATUS DU GATE ARRAY ET
;   OUT (C),C         ; SELECTIONNE L'ANCIENNE CONFIGURATION
;   EXX               ; RETOUR AUX PREMIER REGISTRES
;   EI                ; ACTIVE LES INTERRUPTIONS
;   LD SP,(SSTACK)    ; POINTEUR VERS ANCIENNE PILE
;   RET               ; CA-Y-EST !
;
;   DEFS 10           ; ESPACE POUR PILE LOCALE
SSTACK: DEFS 2         ; ICI SAUVEGARDE DE L'ANCIEN POINTEUR
;   END

```

Parce-que les registres alternes de BC contiennent toujours le Gate Array Status (particularite des CPC), l'activation de la carte est tres facile. L'utilisation des registres alternes des CPC par contre, est toujours un peu delicate (voir le manuel du Firmware).

Si vous voulez recuperer les donnees du bloc inferieur de la banque 0, le processus n'est pas vraiment plus dur, mais il faut quand meme reflechir sur le fait que, lorsque la ROM du Firmware est activee, les donnees sont ecrites dans la RAM, mais lues de la ROM. Quand on veut lire de la RAM, la ROM doit d'abord etre desactivee. Il faut alors remplacer la partie du programme en haut entre les lignes .-.-. par les lignes suivantes:

```

;.....
;
;   PUSH BC           ; SAUVEGARDE LE GATE ARRAY STATUS
;   SET 2,C           ;
;   OUT (C),C         ; DESACTIVE LA ROM INFERIEURE
;   RES 6,C           ; SELECTIONNE BLOC INFERIEUR
;   SET 5,C           ; ACTIVE LA RAM EXTERNE
;   OUT (C),C         ;
;
;
; Remarque: Desactiver la ROM et activer la RAM externe ne peut
; normalement pas etre fait par le meme OUT, parce-que
; le bit 6 n'active pas un Gate Array Register interne
; et alors est ignore par le OUT du Gate Array. Il faut
; donc modifier d'abord la configuration de la ROM et
; puis la configuration de la RAM.
;
;
; MAINTENANT LA RAM EXTERNE EST ACTIVEE: LES DONNEES PEUVENT ETRE
; TRANSFEREES
;
;

```

```
LD HL,0      ; POINTEUR VERS LES DONNEES
LD DE,TAMP   ; ADRESSE DESTINATION = TAMPON
LD BC,128   ; TRANSFERER P.EX. 128 OCTETS
LDIR        ; TRANSFERE BLOC
```

.....

## 2. Stockage de donnees dans le bloc superieur (8000H..FFFFH)

Le stockage de donnees dans un bloc superieur est entierement analogue au stockage dans un bloc inferieur, seules les routines et bien sur la pile doivent se trouver en dessous de 8000H. En lisant des donnees, il faut faire attention de desactiver la ROM superieure (la ROM superieure est geree par le troisieme octet du port du Gate Array - voir manuel du Firmware).

```
-----
: Remarque importante:
: =====
: On pourrait bien ecrire sur le Blocselect Port (=Gate Array
: Port 7FxxH) sans d'abord sauter vers les registres alternes.
: N'oubliez pas que le Gate Array Status peut etre change, ce
: qui normalement est suivi par un 'crash'.
: Pour la meme raison les exemples utilisent les commandes de
: manipulation de bit standard au Z80, parce que ces commandes
: n'interferent pas avec le Gate Array Status. Specialement
: les commandes AND, OR, XOR etc. doivent etre traitees avec
: precaution.
: -----
```